NAG Fortran Library Routine Document

E02ACF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E02ACF calculates a minimax polynomial fit to a set of data points.

2 Specification

```
SUBROUTINE E02ACF(X, Y, N, A, M1, REF)
INTEGER            N, M1
real               X(N), Y(N), A(M1), REF
```

3 Description

Given a set of data points (x_i, y_i), for $i = 1, 2, \ldots, n$, this routine uses the exchange algorithm to compute an mth-order polynomial

$$\mathrm{P}(x) = a_1 + a_2 x + a_3 x^2 + \cdots + a_{m+1} x^m$$

such that $\max_i 2|\mathrm{P}(x_i) - y_i|$ is a minimum.

The routine also returns a number whose absolute value is the final reference deviation (see Section 6). The routine is an adaptation of Boothroyd (1967).

4 References

Stieffel E (1959) Numerical methods of Tchebycheff approximation *On Numerical Approximation* (ed R E Langer) 217–232 University of Wisconsin Press

Boothroyd J B (1967) Algorithm 318 *Comm. ACM* **10** 801

5 Parameters

1: X(N) – ***real*** array *Input*

On entry: the values of the x co-ordinates, x_i, for $i = 1, 2, \ldots, n$.

Constraint: $x_1 < x_2 < \cdots < x_n$.

2: Y(N) – ***real*** array *Input*

On entry: the values of the y co-ordinates, y_i, for $i = 1, 2, \ldots, n$.

3: N – INTEGER *Input*

On entry: the number n of data points.

4: A(M1) – ***real*** array *Output*

On exit: the coefficients a_i of the final polynomial, for $i = 1, 2, \ldots, m + 1$.

5: M1 – INTEGER *Input*

On entry: $m + 1$, where m is the order of the polynomial to be found.

Constraint: $\mathrm{M1} < \min(\mathrm{N}, 100)$.

6: REF – ***real*** *Output*

On exit: the final reference deviation (see Section 6).

6 Error Indicators and Warnings

This routine calls P01ABF internally using the hard fail option to detect the following errors.

IFAIL = 1

$M1 \geq \min(N, 100)$.

IFAIL = 2

The constraint $x_1 < x_2 < \cdots < x_n$ is violated.

With exact arithmetic the algorithm should terminate after a finite number of steps. This need not necessarily be the case using computer arithmetic. Should the routine start cycling then an exit is made with REF given a negative value. This is by no means an indicator that a catastrophic error has occurred and does not preclude useful results being obtained.

The absolute value of REF is the final reference deviation. See Stieffel (1959) for an explanation of this term.

7 Accuracy

This is wholly dependent on the given data points.

8 Further Comments

The time taken by this routine increases with m.

9 Example

The example program calculates a minimax fit with a polynomial of degree 5 to the exponential function evaluated at 21 points over the interval $[0, 1]$. It then prints values of the function and the fitted polynomial.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     E02ACF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          N, M1
      PARAMETER        (N=21,M1=6)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      real             P1
      PARAMETER        (P1=0.1e0)
*     .. Local Scalars ..
      real             REF, S, T, Z
      INTEGER          I, J
*     .. Local Arrays ..
      real             A(M1), X(N), Y(N)
*     .. External Subroutines ..
      EXTERNAL         E02ACF
*     .. Intrinsic Functions ..
      INTRINSIC        EXP, real
*     .. Executable Statements ..
      WRITE (NOUT,*) 'E02ACF Example Program Results'
      DO 20 I = 1, N
```

```
         X(I) = real(I-1)/real(N-1)
         Y(I) = EXP(X(I))
   20 CONTINUE
*
      CALL E02ACF(X,Y,N,A,M1,REF)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) '   Polynomial coefficients'
      WRITE (NOUT,99998) (A(I),I=1,M1)
      WRITE (NOUT,*)
      WRITE (NOUT,99997) '   Reference deviation = ', REF
      WRITE (NOUT,*)
      WRITE (NOUT,*) '  X     exp(X)    Fit     Residual'
      DO 60 J = 1, 11
         Z = real(J-1)*P1
         S = A(M1)
         DO 40 I = M1 - 1, 1, -1
            S = S*Z + A(I)
   40    CONTINUE
         T = EXP(Z)
         WRITE (NOUT,99999) Z, S, T, S - T
   60 CONTINUE
      STOP
*
99999 FORMAT (1X,F5.2,2F9.4,e11.2)
99998 FORMAT (6X,e12.4)
99997 FORMAT (1X,A,e10.2)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 E02ACF Example Program Results

    Polynomial coefficients
        0.1000E+01
        0.1000E+01
        0.4991E+00
        0.1704E+00
        0.3478E-01
        0.1391E-01

    Reference deviation =   0.11E-05

   X     exp(X)    Fit     Residual
  0.00   1.0000   1.0000  -0.11E-05
  0.10   1.1052   1.1052   0.97E-06
  0.20   1.2214   1.2214  -0.74E-06
  0.30   1.3499   1.3499  -0.92E-06
  0.40   1.4918   1.4918   0.30E-06
  0.50   1.6487   1.6487   0.11E-05
  0.60   1.8221   1.8221   0.46E-06
  0.70   2.0138   2.0138  -0.82E-06
  0.80   2.2255   2.2255  -0.84E-06
  0.90   2.4596   2.4596   0.88E-06
  1.00   2.7183   2.7183  -0.11E-05
```